ODBCVerbindungen und DAOZugriffe effektiv debuggen
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Wenn Sie in Ihren Datenbanken per ODBC verknüpfte Tabellen einsetzen und darauf per DAO zugreifen oder Daten aus internen Tabellen per DAO bearbeiten, treten regelmäßig Fehler auf, die es zu bereinigen gilt. In der Regel verwenden Sie dazu eine Konstruktion wie die folgende:

On Error Resume Next
.....
Set db = CurrentDb()
Set rs = db.OpenRecordset( _
         "SELECT * FROM VerknüpfteTabelle")
If Err <> 0 Then
  MsgBox "Fehler: " & CStr(Err.Number) & _
         ", " & Err.Description
  Exit Sub
End If
.....

Über das Error-Objekt erhalten Sie in solchen Fällen lediglich einen Hinweis der Art "Access konnte die Tabelle oder Abfrage 'XYZ' nicht finden...". Damit kennen Sie aber beispielsweise bei einer ODBC-Verbindung die genaue Ursache noch nicht: Ist der Server mit dem Back-End nicht erreichbar? Wurde die richtige Tabelle adressiert? Gab es technische Probleme beim Zugriff auf die Tabelle? Wurde der Zugriff auf die Tabelle wegen fehlender Rechte verweigert? Sie müssen also herausfinden, ob das Problem in Ihrem VBA-Code, in DAO oder in einer unter DAO liegenden Schicht wie zum Beispiel der Jet Engine, dem ODBC-Treiber oder der SQL Server-Authentifizierung liegt. Während sich Probleme im Code, beispielsweise Tippfehler im Namen einer Tabelle oder Objektvariablen vom falschen Datentyp, noch einigermaßen schnell lokalisieren lassen, sind Sie bei Problemen in tiefer liegenden Schichten anscheinend auf sich alleine gestellt.

Wenig bekannt ist, dass Access auch in solchen Fällen Hilfsmittel zur Lokalisierung von Fehlern bereitstellt: Das Error-Objekt ist lediglich das letzte Glied einer Reihe von Hinweisen und gibt Auskunft, warum Access eine Operation schließlich als gescheitert betrachtet hat. Fehler, die auf dem Weg zu den Daten aufgetreten sind, zeigt Access nicht an - andernfalls könnte es zum Beispiel bei einer ODBC-Verbindung durchaus passieren, dass ein halbes Dutzend mehr oder weniger kryptischer Fehlermeldungen beispielsweise von der Jet Engine, dem ODBC-Treibers oder der SQL Server-Authentifizierung zu bestätigen wären, bis schließlich der lapidare Hinweis "Access konnte die Tabelle oder Abfrage 'XYZ' nicht finden..." ausgegeben wird.

Die in unteren Schichten auftretenden Fehler ignoriert Access jedoch nicht einfach, sondern es zeigt sie lediglich nicht an. Festgehalten werden die Fehler in einer Auflistung "Errors" des Objektes "DBEngine". Dabei geht Access so vor, dass für jeden Fehler ein Error-Objekt in die Auflistung "Errors" aufgenommen wird. Das letzte Error-Objekt ist Basis des Access-Laufzeitfehlers, die davor liegenden Error-Objekte beinhalten Informationen beispielsweise der Jet-Engine, des ODBC-Treibers oder einer SQL-Komponente und zwar in der Reihenfolge, in der sie für den jeweiligen Datenzugriff angesprochen werden.

Die folgende Prozedur hilft Ihnen, diese Fehlerhistorie im Falle eines Fehlers beispielsweise mit einer ODBC-Verbindung übersichtlich auszugeben:

Sub ListErrors(blnMsg As Boolean)
  Dim objErr As Error
  Dim strDebug As String
  Dim strMsg As String

  On Error Resume Next
  If DBEngine.Errors.Count = 0 Then
    Debug.Print "Kein Fehler aufgetreten..."
  Else
    For Each objErr In DBEngine.Errors
      With objErr
        strDebug = "Fehler: " & _
                   CStr(.Number) & vbCrLf & _
                   "Ursache: " & _
                   .Description & vbCrLf & _
                   "Quelle: " & _
                   .Source & vbCrLf
        strMsg = strDebug & strMsg
      End With
      Debug.Print strDebug
    Next objErr
    If blnMsg Then MsgBox strMsg
  End If

End Sub

Die Routine schreibt die Fehlerhistorie in das Direkt-/Testfenster der VBA-Entwicklungsumgebung und zeigt sie wahlweise abschließend per "MsgBox" als Zusammenfassung an. Ausgegeben werden jeweils die Fehlernummer (nützlich zum Beispiel für eine Google-Suche), die Klartextmeldung der jeweiligen Komponente sowie die Bezeichnung der Quelle, die den Fehler gemeldet hat (zum Beispiel "DAO.Database" oder der Name einer DLL). In Ihrer Datenbank setzen Sie "ListErrors" dann zum Beispiel so ein:

On Error Resume Next
.....
Set db = CurrentDb()
Set rs = db.OpenRecordset( _
         "SELECT * FROM VerknüpfteTabelle")
If Err <> 0 Then
  ListErrors True
  Exit Sub
End If
.....

